
Foreword 21 

The Foundation for Intelligent Physical Agents (FIPA) is an international organization that is dedicated to promoting the 22 
industry of intelligent agents by openly developing specifications supporting interoperability among agents and agent-23 
based applications. This occurs through open collaboration among its member organizations, which are companies and 24 
universities that are active in the field of agents. FIPA makes the results of its activities available to all interested parties 25 
and intends to contribute its results to the appropriate formal standards bodies where appropriate.  26 

The members of FIPA are individually and collectively committed to open competition in the development of agent-27 
based applications, services and equipment. Membership in FIPA is open to any corporation and individual firm, 28 
partnership, governmental body or international organization without restriction. In particular, members are not bound to 29 
implement or use specific agent-based standards, recommendations and FIPA specifications by virtue of their 30 
participation in FIPA.  31 

The FIPA specifications are developed through direct involvement of the FIPA membership. The status of a 32 
specification can be either Preliminary, Experimental, Standard, Deprecated or Obsolete. More detail about the process 33 
of specification may be found in the FIPA Document Policy [f-out-00000] and the FIPA Specifications Policy [f-out-34 
00003]. A complete overview of the FIPA specifications and their current status may be found on the FIPA Web site. 35 

FIPA is a non-profit association registered in Geneva, Switzerland. As of June 2002, the 56 members of FIPA 36 
represented many countries worldwide. Further information about FIPA as an organization, membership information, 37 
FIPA specifications and upcoming meetings may be found on the FIPA Web site at http://www.fipa.org/. 38 

1 Scope 66 

This specification defines a concrete syntax for the FIPA Semantic Language (SL) content language. This syntax and 67 
its associated semantics are suggested as a candidate content language for use in conjunction with the FIPA Agent 68 
Communication Language (see [FIPA00037]). In particular, the syntax is defined to be a sub-grammar of the very 69 
general s-expression syntax. 70 
 71 

2 Grammar FIPA SL Concrete Syntax 72 

This content language is denoted by the normative constant fipa-sl in the :language parameter of an ACL 73 
message. See Section 6 for an explanation of the used syntactic notation. 74 
 75 
Content                = "(" ContentExpression+ ")". 76 
 77 
ContentExpression      = IdentifyingExpression 78 
                       | ActionExpression  79 
                       | Proposition. 80 
 81 
Proposition            = Wff. 82 
 83 
Wff                    = AtomicFormula 84 
                       | "(" UnaryLogicalOp  Wff ")" 85 
                       | "(" BinaryLogicalOp Wff Wff ")" 86 
                       | "(" Quantifier      Variable Wff ")" 87 
                       | "(" ModalOp         Agent Wff ")" 88 
                       | "(" ActionOp        ActionExpression ")" 89 
                       | "(" ActionOp        ActionExpression Wff ")". 90 
 91 
UnaryLogicalOp         = "not". 92 
 93 
BinaryLogicalOp        = "and" 94 
                       | "or" 95 
                       | "implies" 96 
                       | "equiv". 97 
 98 
AtomicFormula          = PropositionSymbol 99 
                       | "(" BinaryTermOp    TermOrIE TermOrIE ")" 100 
                       | "(" PredicateSymbol TermOrIE+ ")" 101 
                       | "true" 102 
                       | "false". 103 
 104 
BinaryTermOp           = "="                        105 
                       | "result". 106 
 107 
Quantifier             = "forall" 108 
                       | "exists". 109 
 110 
ModalOp                = "B" 111 
                       | "U" 112 
                       | "PG" 113 
                       | "I". 114 
 115 
ActionOp               = "feasible" 116 
                       | "done". 117 
 118 
TermOrIE1               = Term 119 
                       | IdentifyingExpression. 120 
 121 
Term                   = Variable 122 
                       | FunctionalTerm 123 
                       | ActionExpression 124 
                       | Constant 125 
                       | Sequence 126 
                       | Set. 127 
 128 
IdentifyingExpression  = "(" ReferentialOperator TermOrIE Wff ")". 129 
 130 

                                                      
1 Note that this grammar rule is used to group and represent both Terms and Identifying Expressions. 

ReferentialOperator    = "iota" 131 
                       | "any" 132 
                       | "all". 133 
  134 
FunctionalTerm         = "(" FunctionSymbol TermOrIE* ")" 135 
                       | "(" FunctionSymbol Parameter* ")". 136 
 137 
Constant               = NumericalConstant 138 
                       | String 139 
                       | DateTime. 140 
 141 
NumericalConstant      = Integer  142 
                       | Float.  143 
 144 
Variable               = VariableIdentifier.  145 
 146 
ActionExpression       = "(" "action" Agent TermOrIE ")"  147 
                       | "(" "|" ActionExpression ActionExpression ")" 148 
                       | "(" ";" ActionExpression ActionExpression ")". 149 
 150 
PropositionSymbol      = String.  151 
 152 
PredicateSymbol        = String.  153 
 154 
FunctionSymbol         = String.  155 
 156 
Agent                  = TermOrIE.  157 
 158 
Sequence               = "(" "sequence" TermOrIE* ")". 159 
 160 
Set                    = "(" "set" TermOrIE* ")". 161 
 162 
Parameter              = ParameterName ParameterValue.  163 
 164 
ParameterValue      = TermOrIE. 165 
 166 
 167 

2.1 Lexical Definitions 168 

All white space, tabs, carriage returns and line feeds between tokens should be skipped by the lexical analyser. See 169 
Section 6 for an explanation of the used notation.  170 
 171 
String                 = Word 172 
                       | ByteLengthEncodedString 173 
                       | StringLiteral. 174 
 175 
ByteLengthEncodedString = "#" DecimalLiteral+ "\"" <byte sequence>. 176 
 177 
Word                   = [~ "\0x00" - "\0x20", "(", ")", "#", "0" - "9", ":", "-", "?"] 178 
                         [~ "\0x00" - "\0x20", "(", ")"]*. 179 
 180 
ParameterName          = ":" String. 181 
 182 
VariableIdentifier     = "?" String.     183 
 184 
Sign        = [ "+" , "-" ]. 185 
 186 
Integer       = Sign? DecimalLiteral+ 187 
                       | Sign? "0" ["x", "X"] HexLiteral+. 188 
 189 
Dot                    = "." 190 
 191 
Float        = Sign? FloatMantissa FloatExponent? 192 

                       | Sign? DecimalLiteral+ FloatExponent. 193 
 194 
FloatMantissa          = DecimalLiteral+ Dot DecimalLiteral*  195 
                       | DecimalLiteral* Dot DecimalLiteral+. 196 
 197 
FloatExponent          = Exponent Sign? DecimalLiteral+. 198 
 199 
Exponent               = ["e","E"]. 200 
  201 
DecimalLiteral      = ["0" - "9"]. 202 
 203 
HexLiteral        = ["0" - "9", "A" - "F", "a" - "f"]. 204 
  205 
StringLiteral      = "\""( [~ "\""] 206 
                       | "\\\"" )*"\"". 207 
 208 
DateTime               = Sign? Year Month Day "T" Hour Minute 209 
                         Second MilliSecond TypeDesignator?. 210 
 211 
Year         = DecimalLiteral DecimalLiteral DecimalLiteral DecimalLiteral. 212 
 213 
Month        = DecimalLiteral DecimalLiteral. 214 
 215 
Day                    = DecimalLiteral DecimalLiteral. 216 
 217 
Hour                   = DecimalLiteral DecimalLiteral. 218 
 219 
Minute                 = DecimalLiteral DecimalLiteral. 220 
 221 
Second                 = DecimalLiteral DecimalLiteral. 222 
 223 
MilliSecond            = DecimalLiteral DecimalLiteral DecimalLiteral. 224 
 225 
TypeDesignator         = ["a" - "z" , "A" – "Z"]. 226 
 227 

3 Notes on FIPA SL Semantics 228 

This section contains explanatory notes on the intended semantics of the constructs introduced in above. 229 
 230 

3.1 Grammar Entry Point: FIPA SL Content Expression 231 

An FIPA SL content expression may be used as the content of an ACL message. There are three cases: 232 
 233 
• A proposition, which may be assigned a truth value in a given context. Precisely, it is a well-formed formula (Wff) 234 

using the rules described in the Wff production. A proposition is used in the inform communicative act (CA) and 235 
other CAs derived from it. 236 

 237 
• An action, which can be performed. An action may be a single action or a composite action built using the 238 

sequencing and alternative operators. An action is used as a content expression when the act is request and 239 
other CAs derived from it. 240 

 241 
• An identifying reference expression (IRE), which identifies an object in the domain. This is the Referential operator 242 

and is used in the inform-ref macro act and other CAs derived from it. 243 
 244 
Other valid content expressions may result from the composition of the above basic cases. For instance, an action-245 
condition pair (represented by an ActionExpression followed by a Wff) is used in the propose act; an action-246 
condition-reason triplet (represented by an ActionExpression followed by two Wffs) is used in the reject-247 
proposal act. These are used as arguments to some ACL CAs in [FIPA00037]. 248 
 249 

3.2 Well-Formed Formulas 250 

A well-formed formula is constructed from an atomic formula, whose meaning will be determined by the semantics of 251 
the underlying domain representation or recursively by applying one of the construction operators or logical connectives 252 
described in the Wff grammar rule. These are: 253 
 254 
• (not <Wff>) 255 

Negation. The truth value of this expression is false if Wff is true. Otherwise it is true. 256 
 257 
• (and <Wff0> <Wff1>) 258 

Conjunction. This expression is true iff2 well-formed formulae Wff0 and Wff1 are both true, otherwise it is false. 259 
 260 

• (or <Wff0> <Wff1>) 261 
Disjunction. This expression is false iff well-formed formulae Wff0 and Wff1 are both false, otherwise it is true. 262 
 263 

• (implies <Wff0> <Wff1>) 264 
Implication. This expression is true if either Wff0 is false or alternatively if Wff0 is true and Wff1 is true. Otherwise 265 
it is false. The expression corresponds to the standard material implication connective Wff0   Wff1. 266 

 267 
• (equiv <Wff0> <Wff1>) 268 

Equivalence. This expression is true if either Wff0 is true and Wff1 is true, or alternatively if Wff0 is false and 269 
Wff1 is false. Otherwise it is false. 270 

 271 
• (forall <variable> <Wff>) 272 

Universal quantification. The quantified expression is true if Wff is true for every value of value of the quantified 273 
variable. 274 
 275 

• (exists <variable> <Wff>) 276 

                                                      
2 If and only if. 

Existential quantification. The quantified expression is true if there is at least one value for the variable for which 277 
Wff is true. 278 
 279 

• (B <agent> <expression>) 280 
Belief. It is true that agent believes that expression is true. 281 
 282 

• (U <agent> <expression>) 283 
Uncertainty. It is true that agent is uncertain of the truth of expression. Agent neither believes expression 284 
nor its negation, but believes that expression is more likely to be true than its negation. 285 

 286 
• (I <agent> <expression>) 287 

Intention. It is true that agent intends that expression becomes true and will plan to bring it about. 288 
 289 

• (PG <agent> <expression>) 290 
Persistent goal. It is true that agent holds a persistent goal that expression becomes true, but will not 291 
necessarily plan to bring it about. 292 
 293 

• (feasible <ActionExpression> <Wff>) 294 
It is true that ActionExpression (or, equivalently, some event) can take place and just afterwards Wff will be 295 
true. 296 
 297 

• (feasible <ActionExpression>) 298 
Same as (feasible <ActionExpression> true). 299 
 300 

• (done <ActionExpression> <Wff>) 301 
It is true that ActionExpression (or, equivalently, some event) has just taken place and just before that Wff was 302 
true. 303 
 304 

• (done <ActionExpression>) 305 
Same as (done <ActionExpression> true). 306 

 307 

3.3 Atomic Formula 308 

The atomic formula represents an expression which has a truth value in the language of the domain of discourse. Three 309 
forms are defined: 310 
 311 
• A given propositional symbol may be defined in the domain language, which is either true or false, 312 
 313 
• Two terms may or may not be equal under the semantics of the domain language, or, 314 
 315 
• Some predicate is defined over a set of zero or more arguments, each of which is a term. 316 
 317 
The FIPA SL representation does not define a meaning for the symbols in atomic formulae: this is the responsibility of 318 
the domain language representation and ontology. Several forms are defined: 319 
 320 
• true false 321 

These symbols represent the true proposition and the false proposition. 322 
 323 
• (= Term1 Term2) 324 

Term1 and Term2 denote the same object under the semantics of the domain. 325 
   326 
Other predicates may be defined over a set of arguments, each of which is a term, by using the (PredicateSymbol 327 
Term+) production. 328 
  329 

The FIPA SL representation does not define a meaning for other symbols in atomic formulae: this is the responsibility of 330 
the domain language representation and the relative ontology. 331 
 332 

3.4 Terms 333 

Terms are either themselves atomic (constants and variables) or recursively constructed as a functional term in which a 334 
functor is applied to zero or more arguments. Again, FIPA SL only mandates a syntactic form for these terms. With 335 
small number of exceptions (see below), the meanings of the symbols used to define the terms are determined by the 336 
underlying domain representation. 337 
 338 
Note that, as mentioned above, no legal well-formed expression contains a free variable, that is, a variable not declared 339 
in any scope within the expression. Scope introducing formulae are the quantifiers (forall, exists) and the 340 
reference operators iota, any and all. Variables may only denote terms, not well-formed formulae. 341 
 342 

3.5 Referential Operators 343 

3.5.1 Iota 344 

• (iota <term> <formula>) 345 
The iota operator introduces a scope for the given expression (which denotes a term), in which the given 346 
identifier, which would otherwise be free, is defined. An expression containing a free variable is not a well-formed 347 
FIPA SL expression. The expression (iota x (P x)) may be read as “the x such that P [is true] of x”. The iota 348 
operator is a constructor for terms which denote objects in the domain of discourse. 349 
 350 
Notice that, unlike a term, an identifying expression can have different interpretations by different agents because 351 
its formal definition depends on the KB. 352 

 353 
• Formal Definition 354 

A iota expression can only be evaluated with respect to a given theory. Suppose KB is a knowledge base such 355 
that T (KB) is the theory generated from KB by a given reasoning mechanism. Formally, ι(τ, φ)=θτ iff θτ is a term 356 
that belongs to the set Σ={θτ: θφ∈T (KB)} and Σ is a singleton; or ι(τ, φ) is undefined if Σ is not a singleton. In this 357 
definition θ is a most general variable substitution, θτ is the result of applying θ to τ, and θφ is the result of applying 358 
θ to φ. This implies that a failure occurs if no object or more than one object satisfies the condition specified in the 359 
iota operator. 360 
 361 
If ι(τ, φ) is undefined then any term, identifying expression or well-formed formula containing ι(τ, φ) is also 362 
undefined. 363 

 364 
• Example 1 365 

This example depicts an interaction between agent A and B that makes use of the iota operator, where agent A is 366 
supposed to have the following knowledge base KB={P(A), Q(1, A), Q(1, B)}. 367 

 368 
(query-ref  369 
  :sender (agent-identifier :name B) 370 

     :receiver (set (agent-identifier :name A)) 371 
  :content 372 
    "((iota ?x (p ?x)))" 373 
  :language fipa-sl 374 
  :reply-with query1) 375 
 376 
(inform  377 
  :sender (agent-identifier :name A)   378 
  :receiver (set (agent-identifier :name B) 379 
  :content  380 
    " ((= (iota ?x (p ?x)) a)) " 381 
  :language fipa-sl 382 

  :in-reply-to query1) 383 
 384 
The only object that satisfies proposition P(x) is a, therefore, the query-ref message is replied by the inform 385 
message as shown. 386 
 387 

• Example 2 388 
This example shows another successful interaction but more complex than the previous one. 389 
 390 
(query-ref  391 
  :sender (agent-identifier :name B)   392 
  :receiver (set (agent-identifier :name A)) 393 
  :content 394 
    "((iota ?x (q ?x ?y)))" 395 
  :language fipa-sl 396 
  :reply-with query2) 397 
 398 
(inform  399 
  :sender (agent-identifier :name A)   400 
  :receiver (set (agent-identifier :name B)) 401 
  :content 402 
    "((= (iota ?x (q ?x ?y)) 1))" 403 
  :language fipa-sl 404 
  :in-reply-to query2) 405 
 406 
The most general substitutions θ such that θQ(x, y) can be derived from KB are θ1={x/1, y/A} and θ2={x/1, y/B}. 407 
Therefore, the set Σ={θτ: θφ∈T(KB)}={{x/1, y/A}x, {x/1, y/B}x }={1} is a singleton and hence (iota ?x (q ?x ?y)) 408 
represents the object 1. 409 
 410 

• Example 3 411 
Finally, this example shows an unsuccessful interaction using the iota operator. In this case, agent A cannot 412 
evaluate the iota expression and therefore a failure message is returned to agent B 413 
 414 
(query-ref  415 
  :sender (agent-identifier :name B) 416 
  :receiver (set (agent-identifier :name A)) 417 
  :content 418 
    "((iota ?y (q ?x ?y)))" 419 
  :language fipa-sl 420 
  :reply-with query3) 421 
 422 
(failure  423 

     :sender (agent-identifier :name A)   424 
  :receiver (set (agent-identifier :name B)) 425 
  :content 426 
    "((action (agent-identifier :name A) 427 
      (inform-ref  428 
        :sender (agent-identifier :name A)  429 
        :receiver (set (agent-identifier :name B)) 430 
        :content 431 
          \"((iota ?y (q ?x ?y)))\" 432 
        :language fipa-sl 433 
        :in-reply-to query3))" 434 
    more-than-one-answer) 435 
  :language fipa-sl 436 
  :in-reply-to query3) 437 

 438 
The most general substitutions that satisfy Q(x, y) are θ1={x/1, y/a} and θ2={x/1, y/b}, therefore, the set Σ={θτ: 439 
θφ∈T(KB)}={{x/1, y/A}y, {x/1, y/B}y}={A, B}, which is not a singleton. This means that the iota expression used in 440 
this interaction is not defined. 441 

 442 



© 1996-2002 Foundation for Intelligent Physical Agents FIPA SL Content Language 
 

9

3.5.2 Any 443 

• (any <term> <formula>) 444 
The any operator is used to denote any object that satisfies the proposition represented by formula. 445 
 446 
Notice that, unlike a term, an identifying expression can have different interpretations by different agents because 447 
its formal definition depends on the KB. 448 

 449 
• Formal Definition 450 

An any expression can only be evaluated with respect to a given theory. Suppose KB is a knowledge base such 451 
that T(KB) is the theory generated from KB by a given reasoning mechanism. Formally, any(τ, φ)=θτ iff θτ is a term 452 
that belongs to the set Σ={θτ: θφ∈T(KB)}; or any(τ, φ) is undefined if Σ is the empty set. In this definition θ is a most 453 
general variable substitution, θτ is the result of applying θ to τ, and θφ is the result of applying θ to φ. 454 
 455 
If the set Σ is empty then any term, identifying expression or well-formed formula containing any(τ, φ) is undefined. 456 
 457 
If the set Σ is not empty, then for any formula ψ containing any(τ, φ) let ψ' be the formula obtained from ψ by 458 
replacing any(τ, φ) with a variable x (not occurring in ψ) and let s_k be a new Skolem constant. Then ψ is true when 459 
{x/s_k}ψ' element_of T(KB union {τ/s_k}φ), ψ is false when {x/s_k}not(ψ’) element_of T(KB union {τ/s_k}φ), and 460 
otherwise ψ is undefined. 461 
 462 
In other words if ψ contains any(τ, φ), ψ is true if a modified form of ψ obtained by replacing the any expression in it 463 
with a new constant s_k can be inferred based on the assumption that phi holds of s_k. ψ is false if not(ψ) inferred 464 
in a similar way. This definition is needed to avoid the following contradiction: 465 
 466 
(implies 467 
  (and (= Stephen (any ?x (fipa-member ?x))) 468 
       (= Farooq  (any ?x (fipa-member ?x)))) 469 
  (= Stephen Farooq)) 470 

 471 
This definition implies that failures only occur if there are no objects satisfying the condition specified as the second 472 
argument of the any operator. 473 
 474 
If any(τ, φ) is undefined then any term, identifying expression or well-formed formula containing any(τ, φ) is also 475 
undefined. 476 

 477 
• Example 4 478 

Assuming that agent A has the following knowledge base KB={P(A), Q(1, A), Q(1, B)}, this example shows a 479 
successful interaction with agent A using the any operator. 480 

 481 
(query-ref 482 
  :sender (agent-identifier :name B) 483 
  :receiver (set (agent-identifier :name A)) 484 
  :content 485 
    "((any (sequence ?x ?y) (q ?x ?y)))" 486 
  :language fipa-sl 487 
  :reply-with query1) 488 
 489 
(inform 490 
  :sender (agent-identifier :name A) 491 
  :receiver (set (agent-identifier :name B)) 492 
  :content 493 
    "((= (any (sequence ?x ?y) (q ?x ?y)) (sequence 1 a)))" 494 
  :language fipa-sl 495 
  :in-reply-to query1) 496 
 497 

The most general substitutions θ such that θQ(x, y) can be derived from KB are {x/1, y/A} and {x/1, y/B}, therefore 498 
Σ={θSequence(x, y): θQ(x, y)∈T(KB)}={Sequence(1, A), Sequence(1, B)}. Using this set, agent A chooses the first 499 
element of Σ as the appropriate answer to agent B. 500 

 501 
• Example 5 502 

This example shows an unsuccessful interaction with agent A, using the any operator. 503 
 504 
(query-ref  505 
  :sender   (agent-identifier :name B)  506 
  :receiver (set (agent-identifier :name A)) 507 
  :content 508 
    "((any ?x (r ?x)))" 509 
  :language fipa-sl 510 
  :reply-with query2) 511 
 512 
(failure  513 
  :sender (agent-identifier :name A)  514 
  :receiver (set (agent-identifier :name B)) 515 
  :content 516 
    "((action (agent-identifier :name A) 517 
        (inform-ref 518 
          :sender (agent-identifier :name A) 519 
          :receiver (set (agent-identifier :name B)) 520 
          :content  521 
            \"((any ?x (r ?x)))\" 522 
          :language fipa-sl 523 
          :in-reply-to query2)) 524 
    (unknown-predicate r))" 525 
  :language fipa-sl 526 
  :in-reply-to query2) 527 
 528 
Since agent A does not know the r predicate, the answer to the query that had been sent by agent B cannot be 529 
determined, therefore a failure message is sent to agent B from agent A. The failure message specifies the failure’s 530 
reason (that is, unknown-predicate r) 531 

 532 

3.5.3 All 533 

• (all <term> <formula>) 534 
The all operator is used to denote the set of all objects that satisfy the proposition represented by formula. 535 
 536 
Notice that, unlike a term, an identifying expression can have different interpretations by different agents because 537 
its formal definition depends on the KB. 538 

 539 
• Formal Definition 540 

An all expression can only be evaluated with respect to a given theory. Suppose KB is a knowledge base such 541 
that T(KB) is the theory generated from KB by a given reasoning mechanism. Formally, all(τ, φ)={θτ: θφ∈T(KB)}. 542 
Notice that all(τ, φ) may be a singleton or even an empty set. In this definition θ is a most general variable 543 
substitution, θτ is the result of applying θ to τ, and θφ is the result of applying θ to φ. 544 
 545 
If no objects satisfy the condition specified as the second argument of the all operator, then the identifying 546 
expression denotes an empty set. 547 

 548 
• Example 6  549 

Suppose agent A has the following knowledge base KB={P(A), Q(1, A), Q(1, B)}. This example shows a successful 550 
interaction between agent A and B that make use of the all operator. 551 
 552 
(query-ref  553 
  :sender (agent-identifier :name B) 554 

  :receiver (set (agent-identifier :name A)) 555 
  :content 556 
    "((all (sequence ?x ?y) (q ?x ?y)))" 557 
  :language fipa-sl 558 
  :reply-with query1) 559 
 560 
(inform  561 
  :sender (agent-identifier :name A)  562 
  :receiver (set (agent-identifier :name B)) 563 
  :content 564 
    "(( = (all (sequence ?x ?y) (q ?x ?y)) (set(sequence 1 a)(sequence 1 b))))" 565 
  :language fipa-sl 566 
  :in-reply-to query1) 567 
 568 
The set of the most general substitutions θ such that θQ(x, y) can be derived from KB is {{x/1, y/A}, {x/1, y/B}}, 569 
therefore all(Sequence(x, y), Q(x, y))={Sequence(1, A), Sequence(1, B)}. 570 
 571 

• Example 7 572 
Following Example 6, if there is no possible answer to a query making use of the all operator, then the agent 573 
should return the empty set. 574 
 575 
(query-ref  576 
  :sender (agent-identifier :name B)  577 
  :receiver (set (agent-identifier :name A)) 578 
  :content  579 
    "((all ?x (q ?x c)))" 580 
  :language fipa-sl 581 
  :reply-with query2) 582 
 583 
(inform  584 
  :sender (agent-identifier :name A)  585 
  :receiver (set (agent-identifier :name B)) 586 
  :content  587 
    "((= (all ?x (q ?x c))(set)))" 588 
  :language fipa-sl 589 
  :in-reply-to query2) 590 
 591 
Since there is no possible substitution for x such that Q(x, C) can be derived from KB, then all(x, Q(x, c))={}. In this 592 
interaction the term (set) represents the empty set. 593 

 594 

3.6 Functional Terms 595 

A functional term refers to an object via a functional relation (referred by the FunctionSymbol) with other objects (that 596 
is, the terms or parameters), rather than using the direct name of that object, for example, (fatherOf Jesus) rather 597 
than God. 598 
 599 
Two syntactical forms can be used to express a functional term. In the first form the functional symbol is followed by a 600 
list of terms that are the arguments of the function symbol. The semantics of the arguments is position-dependent, for 601 
example, (divide 10 2) where 10 is the dividend and 2 is the divisor. In the second form each argument is preceded 602 
by its name, for example, (divide :dividend 10 :divisor 2). The encoder is required to adopt the following 603 
criteria to select which form to use in order to represent a functional term. The first form, that is, the position-dependent 604 
form, should be used to encode all those functional terms for which the ontology does not specify the names of the 605 
parameters (for example, all the functions of the fipa-agent-management ontology). The second form, that is, the 606 
parameter-name dependent form, must be used to encode all those functional terms for which the ontology does 607 
specify the names of the parameters but not their position (for example, all the object descriptions of the fipa-agent-608 
management ontology). This second form is particularly appropriate to represent descriptions where the function 609 
symbol should be interpreted as the constructor of an object, while the parameters represent the attributes of the object. 610 
 611 
The following is an example of an object, instance of a vehicle class: 612 

 613 
(vehicle 614 
  :colour red 615 
  :max-speed 100 616 
  :owner (Person 617 
    :name Luis 618 
    :nationality Portuguese)) 619 
 620 
Some ontologies may decide to give a description of some concepts only in one or both of these two forms, that is by 621 
specifying, or not, a default order to the arguments of each function in the domain of discourse. How this order is 622 
specified is outside the scope of this specification. 623 
 624 
Functional terms can be constructed by a domain functor applied to zero or more terms. 625 
 626 

3.7 Result Predicate 627 

A common need is to determine the result of performing an action or evaluating a term. To facilitate this operation, a 628 
standard predicate result, of arity two, is introduced to the language. result/2 has the declarative meaning that the 629 
result of evaluating a term, or equivalently of performing an action, encoded by the first argument term, is the second 630 
argument term. However, it is expected that this declarative semantics will be implemented in a more efficient, 631 
operational way in any given FIPA SL interpreter. 632 
 633 
A typical use of the result predicate is with a variable scoped by iota, giving an expression whose meaning is, for 634 
example, ”the x which is the result of agent i performing act”: 635 
 636 
(iota x (result (action i act) x))) 637 
 638 

3.8 Actions and Action Expressions 639 

Action expressions are a special subset of terms. An action itself is introduced by the keyword action and comprises 640 
the agent of the action (that is, an identifier representing the agent performing the action) and a term denoting the action 641 
which is [to be] performed.  642 
 643 
Notice that a specific type of action is an ACL communicative act (CA). When expressed in FIPA SL, syntactically an 644 
ACL communicative act is an action where the agent of the action is the sender of the CA, and the term denotes the 645 
CA including all its parameters where the performative should be used as a function symbol, as referred by the used 646 
ontology. Example 5 includes an example of an ACL CA, encoded as a String, whose content embeds another CA.  647 
 648 
Two operators are used to build terms denoting composite CAs: 649 
 650 
• The sequencing operator (;) denotes a composite act in which the first action (represented by the first operand) is 651 

followed by the second action, and, 652 
 653 
• The alternative operator (|) denotes a composite act in which either the first action occurs, or the second, but not 654 

both. 655 
 656 

3.9 Notes on the Grammar Rules 657 

1. The standard definitions for integers and floating point are assumed. However, due to the necessarily unpredictable 658 
nature of cross-platform dependencies, agents should not make strong assumptions about the precision with which 659 
another agent is able to represent a given numerical value. FIPA SL assumes only 32-bit representations of both 660 
integers and floating point numbers. Agents should not exchange message contents containing numerical values 661 
requiring more than 32 bits to encode precisely, unless some prior arrangement is made to ensure that this is valid. 662 

 663 
2. All keywords are case-insensitive. 664 

 665 
3. A length encoded string is a context sensitive lexical token. Its meaning is as follows: the message envelope of the 666 

token is everything from the leading # to the separator " (inclusive). Between the markers of the message envelope 667 
is a decimal number with at least one digit. This digit then determines that exactly that number of 8-bit bytes are to 668 
be consumed as part of the token, without restriction. It is a lexical error for less than that number of bytes to be 669 
available. 670 

 671 
4. Note that not all implementations of the ACC (see [FIPA00067]) will support the transparent transmission of 8-bit 672 

characters. It is the responsibility of the agent to ensure, by reference to internal API of the ACC, that a given 673 
channel is able to faithfully transmit the chosen message encoding. 674 

 675 
5. Strings encoded in accordance with [ISO2022] may contain characters which are otherwise not permitted in the 676 

definition of Word. These characters are ESC (0x1B), SO (0x0E) and SI (0x0F). This is due to the complexity that 677 
would result from including the full [ISO2022] grammar in the above EBNF description. Hence, despite the basic 678 
description above, a word may contain any well-formed [ISO2022] encoded character, other (representations of) 679 
parentheses, spaces, or the # character. Strings must be enclosed between quote symbols. If the quote symbol 680 
itself needs to be part of the String, then it must be escaped by a \ character. 681 

 682 
6. The format for time tokens is defined in section 3.10. 683 
 684 
7. An agent is represented by its agent-identifier using the standard format from [FIPA00023]. 685 
 686 

3.10 Representation of Time 687 

Time tokens are based on [ISO8601], with extension for relative time and millisecond durations. Time expressions may 688 
be absolute, or relative. Relative times are distinguished by the sign character + or - appearing as the first character in 689 
the token. If no type designator is given, the local time zone is then used. The type designator for UTC is the character 690 
Z; UTC is preferred to prevent time zone ambiguities. Note that years must be encoded in four digits. As an example, 691 
8:30 am on 15th April, 1996 local time would be encoded as: 692 
 693 
19960415T083000000 694 
 695 
The same time in UTC would be: 696 
  697 
19960415T083000000Z 698 
 699 
while one hour, 15 minutes and 35 milliseconds from now would be: 700 
 701 
+00000000T011500035 702 
 703 

4 Reduced Expressivity Subsets of FIPA SL 704 

The FIPA SL definition given above is a very expressive language, but for some agent communication tasks it is 705 
unnecessarily powerful. This expressive power has an implementation cost to the agent and introduces problems of the 706 
decidability of modal logic. To allow simpler agents, or agents performing simple tasks, to do so with minimal 707 
computational burden, this section introduces semantic and syntactic subsets of the full FIPA SL content language for 708 
use by the agent when it is appropriate or desirable to do so. These subsets are defined by the use of profiles, that is, 709 
statements of restriction over the full expressive power of FIPA SL. These profiles are defined in increasing order of 710 
expressivity as FIPA-SL0, FIPA-SL1 and FIPA-SL2. 711 
 712 
Note that these subsets of FIPA SL, with additional ontological commitments (that is, the definition of domain predicates 713 
and constants) are used in other FIPA specifications. 714 
 715 

4.1 FIPA SL0: Minimal Subset 716 

Profile 0 is denoted by the normative constant fipa-sl0 in the language parameter of an ACL message. Profile 0 of 717 
FIPA SL is the minimal subset of the FIPA SL content language. It allows the representation of actions, the 718 
determination of the result a term representing a computation, the completion of an action and simple binary 719 
propositions. The following defines the FIPA SL0 grammar: 720 
 721 
Content            = "(" ContentExpression+ ")". 722 
 723 
ContentExpression  = ActionExpression  724 
                   | Proposition. 725 
 726 
Proposition        = Wff. 727 
 728 
Wff                = AtomicFormula 729 
                   | "(" ActionOp ActionExpression ")". 730 
 731 
AtomicFormula      = PropositionSymbol 732 
                   | "(" "result"        Term Term ")" 733 
                   | "(" PredicateSymbol Term+ ")" 734 
                   | "true" 735 
                   | "false". 736 
 737 
ActionOp           = "done". 738 
 739 
Term               = Constant 740 
              | Set 741 
              | Sequence  742 
              | FunctionalTerm 743 
                   | ActionExpression. 744 
 745 
ActionExpression   = "(" "action" Agent Term ")". 746 
  747 
FunctionalTerm  = "(" FunctionSymbol Term* ")" 748 
                   | "(" FunctionSymbol Parameter* ")". 749 
 750 
Parameter         =  ParameterName ParameterValue. 751 
 752 
ParameterValue  = Term. 753 
 754 
Agent              = Term. 755 
 756 
FunctionSymbol  = String. 757 
 758 
PropositionSymbol  = String. 759 
 760 
PredicateSymbol  = String. 761 

 762 
Constant        = NumericalConstant 763 
                   | String  764 
                   | DateTime.  765 
 766 
Set              = "(" "set" Term* ")". 767 
 768 
Sequence        = "(" "sequence" Term* ")". 769 
 770 
NumericalConstant  = Integer  771 
                   | Float.  772 
 773 
The same lexical definitions described in Section 2.1 apply for FIPA SL0. 774 
 775 

4.2 FIPA SL1: Propositional Form 776 

Profile 1 is denoted by the normative constant fipa-sl1 in the language parameter of an ACL message. Profile 1 of 777 
FIPA SL extends the minimal representational form of FIPA SL0 by adding Boolean connectives to represent 778 
propositional expressions. The following defines the FIPA SL1 grammar: 779 
 780 
Content               = "(" ContentExpression+ ")". 781 
 782 
ContentExpression     = ActionExpression  783 
                      | Proposition. 784 
 785 
Proposition           = Wff. 786 
 787 
Wff                   = AtomicFormula 788 
                 | "(" UnaryLogicalOp  Wff ")" 789 
                 | "(" BinaryLogicalOp Wff Wff ")" 790 
                      | "(" ActionOp        ActionExpression ")". 791 
 792 
UnaryLogicalOp        = "not". 793 
 794 
BinaryLogicalOp     = "and" 795 
                      | "or". 796 
 797 
AtomicFormula         = PropositionSymbol 798 
                      | "(" "result"        Term Term ")" 799 
                      | "(" PredicateSymbol Term+ ")" 800 
                      | "true" 801 
                      | "false". 802 
 803 
ActionOp              = "done". 804 
 805 
Term                  = Constant 806 
                 | Set 807 
                      | Sequence  808 
                 | FunctionalTerm 809 
                      | ActionExpression. 810 
 811 
ActionExpression      = "(" "action" Agent Term ")". 812 
  813 
FunctionalTerm     = "(" FunctionSymbol Term* ")" 814 
                      | "(" FunctionSymbol Parameter* ")". 815 
 816 
Parameter            = ParameterName ParameterValue. 817 
 818 
ParameterValue     = Term. 819 
 820 
Agent                 = Term. 821 
 822 

FunctionSymbol     = String. 823 
 824 
PropositionSymbol     = String. 825 
 826 
PredicateSymbol     = String. 827 
 828 
Constant           = NumericalConstant 829 
                      | String  830 
                      | DateTime. 831 
 832 
Set                 = "(" "set" Term* ")". 833 
 834 
Sequence           = "(" "sequence" Term* ")". 835 
 836 
NumericalConstant     = Integer  837 
                      | Float.  838 
 839 
The same lexical definitions described in Section 2.1 apply for FIPA SL1. 840 
 841 

4.3 FIPA SL2: Decidability Restrictions 842 

Profile 2 is denoted by the normative constant fipa-sl2 in the language parameter of an ACL message. Profile 2 of 843 
FIPA SL allows first order predicate and modal logic, but is restricted to ensure that it must be decidable. Well-known 844 
effective algorithms exist that can derive whether or not an FIPA SL2 Wff is a logical consequence of a set of Wffs (for 845 
instance KSAT and Monadic). The following defines the FIPA SL2 grammar: 846 
 847 
Content               = "(" ContentExpression+ ")". 848 
 849 
ContentExpression     = IdentifyingExpression 850 
                      | ActionExpression  851 
                      | Proposition. 852 
 853 
Proposition           = PrenexExpression. 854 
 855 
Wff                   = AtomicFormula 856 
                      | "(" UnaryLogicalOp  Wff ")" 857 
                      | "(" BinaryLogicalOp Wff Wff ")" 858 
                      | "(" ModalOp         Agent PrenexExpression ")" 859 
                      | "(" ActionOp        ActionExpression ")" 860 
                      | "(" ActionOp        ActionExpression PrenexExpression ")". 861 
 862 
UnaryLogicalOp        = "not". 863 
 864 
BinaryLogicalOp     = "and" 865 
                      | "or" 866 
                      | "implies" 867 
                      | "equiv". 868 
 869 
AtomicFormula         = PropositionSymbol 870 
                      | "(" "="             TermOrIE TermOrIE ")" 871 
                      | "(" "result"        TermOrIE TermOrIE ")" 872 
                      | "(" PredicateSymbol TermOrIE+ ")" 873 
                      | "true" 874 
                      | "false". 875 
 876 
PrenexExpression      = UnivQuantExpression 877 
                      | ExistQuantExpression 878 
                      | Wff. 879 
 880 
UnivQuantExpression   = "(" "forall" Variable Wff ")" 881 
                      | "(" "forall" Variable UnivQuantExpression ")" 882 
                      | "(" "forall" Variable ExistQuantExpression ")". 883 

 884 
ExistQuantExpression  = "(" "exists" Variable Wff ")" 885 
                      | "(" "exists" Variable ExistQuantExpression ")". 886 
 887 
TermOrIE              = Term 888 
                      | IdentifyingExpression. 889 
 890 
Term                  =  Variable 891 
                      | FunctionalTerm 892 
                      | ActionExpression 893 
                      | Constant 894 
                      | Sequence 895 
                      | Set. 896 
  897 
IdentifyingExpression = "(" ReferentialOp TermOrIE Wff ")". 898 
 899 
ReferentialOp     = "iota" 900 
                      | "any" 901 
                      | "all". 902 
 903 
FunctionalTerm     = "(" FunctionSymbol TermOrIE* ")" 904 
                      | "(" FunctionSymbol Parameter* ")". 905 
 906 
Parameter            = ParameterName ParameterValue. 907 
 908 
ParameterValue     = TermOrIE. 909 
 910 
ActionExpression     = "(" "action" Agent TermOrIE ")" 911 
                 | "(" "|" ActionExpression ActionExpression ")" 912 
                 | "(" ";" ActionExpression ActionExpression ")". 913 
 914 
Variable              = VariableIdentifier. 915 
 916 
Agent                 = TermOrIE. 917 
 918 
FunctionSymbol        = String. 919 
 920 
Constant           = NumericalConstant  921 
                      | String  922 
                      | DateTime. 923 
 924 
ModalOp               = "B" 925 
                      | "U" 926 
                      | "PG" 927 
                      | "I". 928 
 929 
ActionOp           = "feasible"  930 
                      | "done". 931 
 932 
PropositionSymbol     = String. 933 
 934 
PredicateSymbol     = String. 935 
 936 
Set                 = "(" "set" TermOrIE* ")". 937 
 938 
Sequence           = "(" "sequence" TermOrIE* ")". 939 
 940 
NumericalConstant      = Integer  941 
                       | Float.  942 
 943 
  944 
The same lexical definitions described in Section 2.1 apply for FIPA SL2. 945 
 946 

The Wff production of FIPA SL2 no longer directly contains the logical quantifiers, but these are treated separately to 947 
ensure only prefixed quantified formulas, such as: 948 
 949 
(forall ?x1 950 
  (forall ?x2  951 
    (exists ?y1 952 
      (exists ?y2  953 
        (Phi ?x1 ?x2 ?y1 ?y2))))) 954 
 955 
Where (Phi ?x1 ?x2 ?y1 ?y2) does not contain any quantifier. 956 
  957 
The grammar of FIPA SL2 still allows for quantifying-in inside modal operators. For example, the following formula is 958 
still admissible under the grammar: 959 
 960 
(forall ?x1  961 
  (or  962 
    (B i (p ?x1)) 963 
    (B j (q ?x1)))) 964 
 965 
It is not clear that formulae of this kind are decidable. However, changing the grammar to express this context 966 
sensitivity would make the EBNF form above essentially unreadable. Thus, the following additional mandatory 967 
constraint is placed on well-formed content expressions using FIPA SL2: Within the scope of an SLModalOperator 968 
only closed formulas are allowed, that is, formulas without free variables. 969 
 970 

6 Informative Annex A — Syntax and Lexical Notation 980 

The syntax is expressed in standard EBNF format. For completeness, the notation is given in Table 2. 981 
 982 

Grammar rule component  Example 
Terminal tokens are enclosed in double quotes "(" 

Non terminals are written as capitalised identifiers Expression 

Square brackets denote an optional construct [ "," OptionalArg ] 

Vertical bar denotes an alternative Integer | Real 

Asterisk denotes zero or more repetitions of the preceding expression Digit * 

Plus denotes one or more repetitions of the preceding expression Alpha + 

Parentheses are used to group expansions ( A | B ) * 

Productions are written with the non-terminal name on the left-hand 
side, expansion on the right-hand side and terminated by a full stop 

AnonTerminal = "an expansion". 

 983 
Table 2: EBNF Rules 984 

 985 
Some slightly different rules apply for the generation of lexical tokens. Lexical tokens use the same notation as above, 986 
with the exceptions noted in Table 3. 987 
 988 

Lexical rule component  Example 
Square brackets enclose a character set ["a", "b", "c"] 

Dash in a character set denotes a range ["a" - "z"] 

Tilde denotes the complement of a character set if it is the first 
character 

[~ "(", ")"] 

Post-fix question-mark operator denotes that the preceding lexical 
expression is optional (may appear zero or one times) 

["0" - "9"]? ["0" - "9"] 

 989 
Table 3: Lexical Rules 990 

 991 

7 Informative Annex B — ChangeLog 992 

7.1 2002/11/01 - version H by TC X2S 993 

Entire document: Fixed bugs in the examples, by adding quotes and converting symbols into lower case 994 
Entire document: Added new non-terminal symbol TermOrIE and replaced all occurrences of Term with 995 

TermOrIE 996 
Page 2, line 72: Added symbol identifying fipa-sl content language 997 
Page 2, lines 104-112: Removed superfluous binary term operators 998 
Page 3, lines 139-149: Removed superfluous functional term operators 999 
Page 3, lines 180-184: Removed superfluous arithmetic operators 1000 
Page 4, line 224: Added optional Sign symbol to represent relative time 1001 
Pages 6, lines 342-373: Removed description of superfluous equality operators 1002 
Page 8, line 398: Added note on interpretation of iota identifying expression 1003 
Page 8, line 406: Added note on interpretation of iota identifying expression 1004 
Page 9, line 488 : Added note on interpretation of any identifying expression 1005 
Page 9, line 494: Improved the definition of any identifying expression 1006 
Page 9, line 497: Improved the definition of any identifying expression 1007 
Page 10, line 556: Added note on interpretation of all identifying expression 1008 
Page 11, line 619: Added requirement on encoding functional terms 1009 
Page 12, line 639: Removed Table 1 on description of superfluous functional operators 1010 
Page 12, lines 660-662: Removed ambiguity in representing communicative acts in SL 1011 
Page 12, line 664: Added description of the actor of an ACL Message 1012 
Page 13, lines 672-674: Removed section on agent identifiers 1013 
Page 13, lines 375-380: Extended the section on Numerical Constants to incorporate more details on Grammar Rules 1014 
Page 13, lines 682-692 : Extended the section on Date and Time Constants to add a description of relative time 1015 
 1016 

7.2 2002/12/03 - version I by FIPA Architecture Board 1017 

Entire document: Promoted to Standard status 1018 
 1019 


